MLC Lab Visit

Mth 351 May 18 2001 Maple 6

Bent E. Petersen

Filename: 351s2001_lab_visit.mws

Some of the commands below may not work correctly in Maple 5 and earlier. Usually there are other ways to achieve the same effect.

The workstations in the MLC lab are PCs running Windows NT 4.0. In order to use the machines you must have an ORST (a.k.a. myORST) account. When you logon to a machine in the MLC lab your ORST directory on the ORST server will be visible as drive Z: This is where you should keep your personal files. Then they will be available from any PC in the MLC lab (and many other labs) when you login.

Login

The machines in the lab are normally left on, but the monitors may be turned off. If the monitor is off, then switch it on. Next press the Ctrl-Alt-Delete keys simultaneously. You should get a login prompt. Enter your ORST user name and press the Tab key (not the Enter key). Then enter your ORST password and press the Enter key.

Next you may see a message about a slow network connection. This message is bogus. Ignore it.

Next you may see a question about a default Novell server. Just answer "none" unless you have a reason to answer otherwise. This question should never appear again.

To start Maple (or Matlab, or Mathematica, ...) select the Start button (lower left corner of the screen), then Programs from the menu, etc. If you don't know the appropriate steps here ask for help. Writing all this out in detail produces an incredibly dull document.

Logout

When you are done with your session save your work, shutdown the software you were using and should logout. To logout select the Start button (this is a very strange Windows idiom), and then select "Shut Down ..." and finally select "Logon as another user."

Another way to logout is to press Ctrl-Alt-Delete. A menu will appear. Select logoff. That is the simplest way.

Do not select Restart or Shutdown unless you have a reason to do so and do not shut off the PC. There is no harm in shutting off the PC, but doing so causes the next user to have a long wait while the machine reboots. You may turn off the monitor if you wish. That will save power and will reduce the load on the air-conditioner.

Note: If you do not logout you leave your account open for the next person to come along. That person will have access to your personal files on the ORST server. Do not forget to logout!

If you plan to leave the lab, even just for a few minutes, save your work and logoff. When you return and logon, even to a different machine, your work will be available. Do not select "Lock Workstation." If you do, someone else wishing to use the workstation may power-cycle it in order to gain access and some of your work may be lost as a result. The same comments apply to relying on a password protected screensaver. Don't do it. Save your work and logoff. You have no claim on any workstation if you are not physically present.

The Worksheet

When you are using Maple in a window environment it is possible to move around on the worksheet by left-clicking the mouse. As a result, commands may end up being executed in a nonlinear order. This can cause some confusion, since there is no visual clue. One way to fix a mess is to have Maple re-execute the whole worksheet (look on the Edit menu). This works best if old expressions are cleaned up first, so it is a good idea to start each worksheet with the command restart; You do not need to do so of course ....

```
> restart;
```

Maple commands are executed by pressing the Enter key when the mouse cursor is in the line containing the commands.

Note each Maple command must be terminated by a colon or a semicolon (except help commands preceded by a question mark). You can spread the command over several lines by postponing the terminating colon or semicolon. You simply move to a new line by pressing Enter. Maple will chatter at you when you move to a new line in this manner if the previous command is unterminated. Ignore it, but keep in mind a command will not be executed before it is properly terminated.

You can also stack up several commands on one line by terminating them individually with colons or semicolons. The effect of the colon is to suppress output from the corresponding command, though the command is still carried out. All the commands on a line are executed when you press the Enter key.

Maple has two ditto operators, % and %%. The value of % is the previous evaluated expression, the value of %% is the one before that. Since the Worksheet commands may be executed in any order, the ditto operators can cause a lot of confusion. It is probably best to restrict them to the same line as the expressions they refer to. Here is a silly example, which also demonstrates the assignment operator.

Here's a useful fact: You can open a new command line below the current one by pressing Ctrl-J, or above the current line, by pressing Ctrl-K. This is pretty useful when you realize you omitted something at a certain step.

```
> a:=5: b:=4: %%; %%; %;
```

$$5$$
$$4$$
$$4$$

You can also unassign variables. Right now a is 5. That would cause problems if we want to use a as a dummy variable of integration!

```
> unassign('a','b'); a; b;
```

$$a$$
$$b$$

You can pass any number of variables to the unassign() command.

A simpler way to unassign one variable is to assign it its name extracted by single quotes (this is a Maple idiom)

```
> a:=5; a:='a': a;
```

$$a := 5$$
$$a$$

This is quite convenient, but sometimes the single quotes are hard to find on the keyboard and even harder to see on the monitor. Thus, even though it is more typing you may prefer to use the evaluate to a name function evaln() since it does not require the pesky single quotes.

```
> a:=5; b:=4;
```

$$a := 5$$
$$b := 4$$

```
> unassign(evaln(a),evaln(b)); a; b;
```

$$a$$
$$b$$

Unfortunately, you can pass only one expression to evaln(), since it returns only one name.

Maple has builtin constants

```
> Pi; evalf(Pi); I; I^2;
```

$$\pi$$
$$3.141592654$$

I

-1

Note the upper case letters. If you enter pi you will just get the Greek letter pi, not the real number pi. By the way, the evalf() function takes a second parameter specifiying the precision in decimal digits.

```
> evalf(Pi,40);
```

3.141592653589793238462643383279502884197

You can also set the precision by assigning a value to Digits (the default is 10). Maple usually does exact calculations, but when floating point numbers are involved then Digits sets the precision. Here's an amusing example

```
> Digits:=4: convert(evalf(Pi),`rational`);
```

$$\frac{22}{7}$$

The conversion to a rational number makes use of Digits, rather than any precision specified in the evalf() command. You can easily find other rational approximations to pi

```
> Digits:=8: convert(evalf(Pi),rational);
```

$$\frac{355}{113}$$

Note this time I omitted the backticks on the word "rational." Most of the time you do not need them, but if you have a variable called "rational" you need the backticks to ensure that you pass a literal string to Maple's convert() function, rather than the value of your variable "rational."

Let's set Digits back to its default.

```
> Digits:=10:
```

Let's look a bit at symbolic manipulations now. Maple distinguishes between functions and expressions. Here's one way to define a function:

```
> f:=x->sin(3*x+x^2);
```

$$f := x \rightarrow \sin(3\,x + x^2)$$

We can also define an expression:

```
> g:=sin(3*x+x^2);
```

$$g := \sin(3\,x + x^2)$$

Both of the examples above assume that x has not already been assigned a value. It needs to be an unassigned variable. In the definition of f the x is a dummy variable, a place marker. In g however, it is part of the expression, and one can refer to it.

To evaluate a function we use the usual function convention. To evaluate an expression one generally uses the subs() command (though it has other subtle uses).

```
> f(1); subs(x=1,g);
```

$$\sin(4)$$

$$\sin(4)$$

Note the subs() command above does not assign a value to x.

An expression can also be evaluated by using the eval() command, but do check help to make sure you don't have any surprises in more complicated situations. The commands eval() and subs() work in quite different ways. In the simple case that we illustrated here eval() is actually the preferred command to use.

```
> eval(g,x=1);
```

$$\sin(4)$$

Note the eval() command above does not assign a value to x.

We can convert an expression into a function by using the unapply() command

```
> h:=unapply(g,x);
```

$$h := x \rightarrow \sin(3\,x + x^2)$$

You can think of unapply() as turning the indicated variable(s) into dummy variables or place markers.

Some Maple commands work on expressions, some work on functions, and some on both. For example, here are the derivatives of f and g.

```
> D(f); diff(g,x);
```

$$x \rightarrow \cos(3\,x + x^2)\,(3 + 2\,x)$$

$$\cos(3\,x + x^2)\,(3 + 2\,x)$$

Second derivatives are no problem

```
> D(D(f)); diff(g,x,x);
```

$$x \to -\sin(3\,x+x^2)\,(3+2\,x)^2+2\cos(3\,x+x^2)$$

$$-\sin(3\,x+x^2)\,(3+2\,x)^2+2\cos(3\,x+x^2)$$

but this notation can get out hand. Fortunately there is an alternative! Here are the fourth derivatives as an illustration:

```
> (D@@4)(f); diff(g,x$4);
```

$$x \to \sin(3\,x+x^2)\,(3+2\,x)^4-12\cos(3\,x+x^2)\,(3+2\,x)^2-12\sin(3\,x+x^2)$$

$$\sin(3\,x+x^2)\,(3+2\,x)^4-12\cos(3\,x+x^2)\,(3+2\,x)^2-12\sin(3\,x+x^2)$$

Partial derivatives of expressions are also easily computed:

```
> diff(x/(x^2+y^2),x$3,y);
```

$$-288\,\frac{x^2\,y}{(x^2+y^2)^4}+\frac{24\,y}{(x^2+y^2)^3}+\frac{384\,x^4\,y}{(x^2+y^2)^5}$$

There is an inert version Diff() of diff(). An inert function returns unevaluated. That may seem strange, but sometimes one can save time by postponing evaluation, or one can prevent Maple from attempting a calculation that will fail in at present, but can be caried out later in special cases. Unevaluated expressions may be evaluated by using the command value(), thought there are other ways.

Inert functions, together with the ditto operator can be used to get nicely typeset expressions. See if you can sort out the following:

```
> Diff(x/(x^2+y^2),x$3,y): %=value(%);
```

$$\frac{\partial^4}{\partial y\,\partial x^3}\,\frac{x}{x^2+y^2}=-288\,\frac{x^2\,y}{(x^2+y^2)^4}+\frac{24\,y}{(x^2+y^2)^3}+\frac{384\,x^4\,y}{(x^2+y^2)^5}$$

As a final general example let’s bring back some fond memories from calculus - the problem of integration. Here’s on example to get you started: Once again I use postponed evaluation to get a nicely typeset equation. You don’t need to do such trickery, of course, but it’s nice to know how.

```
> Int(1/(1+x^4),x): % = value(%);
```

$$\int \frac{1}{1+x^4}\,dx=\frac{1}{8}\sqrt{2}\,\ln\left(\frac{x^2+x\sqrt{2}+1}{x^2-x\sqrt{2}+1}\right)+\frac{1}{4}\sqrt{2}\,\arctan(x\sqrt{2}+1)+\frac{1}{4}\sqrt{2}\,\arctan(x\sqrt{2}-1)$$

You can obtain the same effect by writing

```
> Int(1/(1+x^4),x) = int(1/(1+x^4),x);
```

$$\int \frac{1}{1+x^4}\,dx = \frac{1}{8}\sqrt{2}\,\ln\left(\frac{x^2+x\sqrt{2}+1}{x^2-x\sqrt{2}+1}\right)+\frac{1}{4}\sqrt{2}\,\arctan(x\sqrt{2}+1)+\frac{1}{4}\sqrt{2}\,\arctan(x\sqrt{2}-1)$$

if you don't mind writing the integrand twice. If you are just interested in evaluating the integral then you can dispense with all the typesetting niceties:

```
> int(1/(1+x^4),x);
```

$$\frac{1}{8}\sqrt{2}\,\ln\left(\frac{x^2+x\sqrt{2}+1}{x^2-x\sqrt{2}+1}\right)+\frac{1}{4}\sqrt{2}\,\arctan(x\sqrt{2}+1)+\frac{1}{4}\sqrt{2}\,\arctan(x\sqrt{2}-1)$$

Naturally definite integrals are possible too.

```
> Int(2*x^2*log(x)^3+x^3*log(x),x=1..2): %=value(%);
```

$$\int_1^2 2\,x^2\ln(x)^3+x^3\ln(x)\,dx = -\frac{16}{3}\ln(2)^2+\frac{68}{9}\ln(2)-\frac{853}{432}+\frac{16}{3}\ln(2)^3$$

If you want a floating point number you can simply use evalf(), but there is a subtle and important difference depending on how you do it.

```
> a:=int(2*x^2*log(x)^3+x^3*log(x),x=1..2): evalf(a,16);
```

$$2.476290396904212$$

```
> evalf(Int(2*x^2*log(x)^3+x^3*log(x),x=1..2),16);
```

$$2.476290396904210$$

In the first case we assign the symbolic expression for the integral to a and then evaluate that expression. In the second example, Maple detects that we want a numeric result and evaluates the integral numerically without first trying to obtain a symbolic solution. This is important. For example

```
> int(arctan(x)/log(x),x=Pi/8..Pi/4); evalf(%);
```

$$\int_{1/8\,\pi}^{1/4\,\pi} \frac{\arctan(x)}{\ln(x)}\,dx$$

$$-.4623890373$$

```
> evalf(Int(arctan(x)/log(x),x=Pi/8..Pi/4));
```

$$-.4623890373$$

Here, in the first case, Maple decided after a while that it can not return a symbolic value for the integral and so returned it unevaluated. Then evalf() called a numeric quadrature rule to get an answer. In the second case however, Maple wasted no time trying to find a nonexistent symbolic solution, but instead used a numeric quadrature method. This is an important use of inert functions. You can grow noticably older waiting for a symbolic solution to a complex problem.

There are refinements. For example, you can specify what quadrature method to use. Enter the command ?int[numeric] for more information.

Some Plots

Functions and expressions can be plotted. There are numerous plot variations. Check the help facility, ?plot, for details.

```
> f:=x->sin(1/x); g:=sin(1/x);
```

$$f := x \to \sin\left(\frac{1}{x}\right)$$

$$g := \sin\left(\frac{1}{x}\right)$$

```
> plot(f,0..1,numpoints=200,title="Plotting a function");
```




```
> plot(g,x=0..1,numpoints=200,title="Plotting an expression");
```



We can convert a function into an expression simply by evaluating it, so one can also do

```
> plot(f(x),x=0..1):
```

I supressed the output, since you probably don't want to see a third copy of the same graph.

You can also plot anonymous functions, or expressions, that is, plot them without first assigning them to a variable:

```
> plot(x->x+sin(x),0..4*Pi);
```




```
> plot(x+sin(x),x=0..4*Pi);
```



Taylor Series and Polynomials

We have studied Taylor and interpolation polynomials in class. Maple supports both. Actually the

Taylor polynomial is a special case of interpolation, with all the nodes equal. but Maple's interpolation routine requires distinct nodes (we can get around that restriction by using limits).

Let's start with an example of Taylor polynomials:

```
> expr:=taylor(exp(2*sin(x)),x=0,10);
```

$$expr := 1 + 2\,x + 2\,x^2 + x^3 - \frac{23}{60}\,x^5 - \frac{4}{15}\,x^6 - \frac{19}{360}\,x^7 + \frac{2}{45}\,x^8 + \frac{7057}{181440}\,x^9 + \mathrm{O}(x^{10})$$

Note taylor() works on expressions. The second argument specifies the center. The third argument specifies the order of the terms omitted. Parameters may be included:

```
> a:=evaln(a);
```

$$a := a$$

```
> expr:=taylor(exp(a*sin(x)),x=0,5);
```

$$expr := 1 + a\,x + \frac{1}{2}\,a^2\,x^2 + \left(-\frac{1}{6}\,a + \frac{1}{6}\,a^3\right)x^3 + \left(-\frac{1}{6}\,a^2 + \frac{1}{24}\,a^4\right)x^4 + \mathrm{O}(x^5)$$

Note I unevaluated a first, because we left it assigned to some number above. If I had not unevaluated it then Maple would have substituted the value of a in this expression.

The data type returned by taylor() is a series, not a polynomial. If you want a polynomial to play with you need to do a conversion:

```
> taylor(tan(x),x=0,10): p:=convert(%,polynom);
```

$$p := x + \frac{1}{3}\,x^3 + \frac{2}{15}\,x^5 + \frac{17}{315}\,x^7 + \frac{62}{2835}\,x^9$$

You need backticks if you have used polynom as a vriable name.

You can specify a different center, even a symbolic one

```
> taylor(exp(x),x=c,4): pc:=convert(%,polynom);
```

$$pc := \mathbf{e}^c + \mathbf{e}^c\,(x-c) + \frac{1}{2}\,\mathbf{e}^c\,(x-c)^2 + \frac{1}{6}\,\mathbf{e}^c\,(x-c)^3$$

Interpolation Polynomials

Maple provides a builtin command for computing interpolation polynomials.

```
> q1:=interp([1,3,4,2],[2,1,3,1],x);
```

$$q1 := \frac{1}{6}x^3 - \frac{1}{2}x^2 - \frac{2}{3}x + 3$$

The first parameter we pass to interp() is the list of (distinct) abscissas, the second is the list of ordinates and the third is a name, the name for the variable to be used in the polynomial.

If you want a polynomial function rather than a polynomial expression in some variable, you can use unapply():

```
> q2:=unapply(interp([1,3,4,2],[2,1,3,1],x),x);
```

$$q2 := x \to \frac{1}{6}x^3 - \frac{1}{2}x^2 - \frac{2}{3}x + 3$$

Let's check that it worked:

```
> q2(1); q2(3); q2(4); q2(2);
```

$$2$$
$$1$$
$$3$$
$$1$$

If you have a list of points you want to interpolate you can extract the abscissas and ordinates by using the op() command (it lists the operands in its argument):

```
> L:=[ [1,2], [2,-1], [3,-2], [4,1], [5,7], [6,6], [7,5] ];
```

$$L := [[1, 2], [2, -1], [3, -2], [4, 1], [5, 7], [6, 6], [7, 5]]$$

We start by declaring two empty lists, XX and YY, and then push the abscissas on XX and the ordinates on YY:

```
> XX:=[]: YY:=[]: for ptr in L do XX:=[op(XX),ptr[1]];
  YY:=[op(YY),ptr[2]]; od:
```

Before we use XX and YY let's check that they look alright

```
> XX; YY;
```

$$[1, 2, 3, 4, 5, 6, 7]$$
$$[2, -1, -2, 1, 7, 6, 5]$$

```
> p3:=interp(XX,YY,x);
```

$$p3 := \frac{2}{45}x^6 - \frac{59}{60}x^5 + \frac{605}{72}x^4 - \frac{106}{3}x^3 + \frac{28099}{360}x^2 - \frac{5291}{60}x + 40$$

```
> plot(p3,x=1..7,title="p3");
```



A convenient way to construct an interpolation polynomial for a function is to use the map() command to evaluate the function at each abscissa. Let's consider the sine function on [0,4]:

```
> XX:=[0,1/2,1,3/2,2,5/2,3,7/2,4];
```

$$XX := \left[0, \frac{1}{2}, 1, \frac{3}{2}, 2, \frac{5}{2}, 3, \frac{7}{2}, 4\right]$$

```
> YY:=evalf(map(sin,XX));
```

YY := [0., .4794255386, .8414709848, .9974949866, .9092974268, .5984721441, .1411200081, -.3507832277, -.7568024953]

Here we used evalf() to force (approximate) evaluation of the sine. Otherwise we will get an (painfully) exact answer. Try it.

```
> ps:=interp(XX,YY,t);
```

$$ps := .00002074516762\, t^8 - .0002575581726\, t^7 + 1.000090277\, t + .0000235057535\, t^6 - .00045081425\, t^2 + .008609497669\, t^5 - .1658254022\, t^3 - .000739266491\, t^4$$

```
> plot([ps,sin(t)],t=0..6.5,title="Interpolation of
  sin(t)",color=[red,blue]);
```



Note the previous example shows one way of plotting two functions on one graph.

Interpolating Spline

Maple computes splines of all degrees - check the help. Here we will look only at linear and (natural) cubic splines. A linear spline is just a piecewise linear function. The parameters are much the same as for interp(), but the abscissas must be in increasing order.

```
> XX:=[1,2,5/2,3,13/4,15/4,5];
```

$$XX := \left[1, 2, \frac{5}{2}, 3, \frac{13}{4}, \frac{15}{4}, 5 \right]$$

```
> YY:=[1,1,2,1,-1,-1,3];
```

$$YY := [1, 1, 2, 1, -1, -1, 3]$$

```
> sp1:=spline(XX,YY,x,linear);
```

$$sp1 := \begin{cases} 1 & x < 2 \\ -3 + 2\,x & x < \frac{5}{2} \\ 7 - 2\,x & x < 3 \\ 25 - 8\,x & x < \frac{13}{4} \\ -1 & x < \frac{15}{4} \\ -13 + \frac{16}{5}\,x & otherwise \end{cases}$$

```
> sp3:=spline(XX,YY,x,cubic);
```

$$sp3 := \begin{cases} 1 + \frac{41008}{24395}\,x - \frac{61512}{24395}\,x^2 + \frac{20504}{24395}\,x^3 & x < 2 \\ \frac{184719}{4879} - \frac{1307792}{24395}\,x + \frac{612888}{24395}\,x^2 - \frac{13128}{3485}\,x^3 & x < \frac{5}{2} \\ \frac{450319}{4879} - \frac{2901392}{24395}\,x + \frac{1250328}{24395}\,x^2 - \frac{176888}{24395}\,x^3 & x < 3 \\ -\frac{4412539}{3485} + \frac{30237976}{24395}\,x - \frac{9796128}{24395}\,x^2 + \frac{1050496}{24395}\,x^3 & x < \frac{13}{4} \\ \frac{3062588}{4879} - \frac{12408836}{24395}\,x + \frac{3325968}{24395}\,x^2 - \frac{59072}{4879}\,x^3 & x < \frac{15}{4} \\ -\frac{43627}{4879} + \frac{16024}{24395}\,x + \frac{12672}{24395}\,x^2 - \frac{4224}{121975}\,x^3 & otherwise \end{cases}$$

```
> plot([sp1,sp3],x=1..5,color=[red,blue],thickness=2,numpoints=200,t
  itle="piecewise linear and cubic spline interpolation");
```



You can see how the cubic spline smoothens out the graph without introducing too much oscillation. If we compare the piecewise linear spline and the interpolation polynomial we see unreasonable oscillation unsupported by the data:

```
> pp:=interp(XX,YY,x);
```

$$pp := -\frac{128488}{51975}x^6 + \frac{730628}{17325}x^5 - \frac{29835503}{103950}x^4 + \frac{68943997}{69300}x^3 - \frac{95898871}{51975}x^2 + \frac{7969177}{4620}x - \frac{41341}{66}$$

```
> plot([sp1,pp],x=1..5,color=[red,blue],thickness=2,numpoints=200,title="piecewise linear and polynomial interpolation");
```



Note the vertical scales are different in the two graphs. We can plot all three function in one graph for a more convincing demonstartion of how well the cubic spline follows the piecewise interpolation.

```
> plot([sp1,sp3,pp],x=1..5,-3..9,color=[red,blue,black],thickness=2,
  numpoints=200,title="red=piecewise linear, blue=cubic spline,
  black=interpolation polynomial");
```



Note how I restricted the vertical range so we would be able to see the details (otherwise the piecewise linear and the cubic spline just about merge on the graph).

Trapezoidal and Simpson's Rule

Maple has numerous high-power quadrature methods built in, but if one simply wants to experiment with the trapezoidal rule or Simpson's rule, these are available in the student package, accessed through the command with(student).

It is also fairly easy to roll your own, even to write high order Newton-Cotes methods, if you wish. There are some example on my web page. For now, let's use the student package.

```
> with(student):
> trapezoid(f(x),x=a..b,6);
```

$$\frac{1}{2}\left(\frac{1}{6}b-\frac{1}{6}a\right)\left(\sin\left(\frac{1}{a}\right)+2\left(\sum_{i=1}^{5}\sin\left(\frac{1}{a+i\left(\frac{1}{6}b-\frac{1}{6}a\right)}\right)\right)+\sin\left(\frac{1}{b}\right)\right)$$

```
> simpson(f(x),x=a..b,6);
```

$$\frac{1}{3}\left(\frac{1}{6}b-\frac{1}{6}a\right)$$

$$\left(\sin\left(\frac{1}{a}\right)+\sin\left(\frac{1}{b}\right)+4\left(\sum_{i=1}^{3}\sin\left(\frac{1}{a+(2\,i-1)\left(\frac{1}{6}b-\frac{1}{6}a\right)}\right)\right)+2\left(\sum_{i=1}^{2}\sin\left(\frac{1}{a+2\,i\left(\frac{1}{6}b-\frac{1}{6}a\right)}\right)\right)\right)$$

Let's try an actual function, say exp(x)*cos(x).

```
> trapezoid(exp(x)*cos(x),x=0..3,12): test:=evalf(%);
```

$$test := -9.148761413$$

```
> simpson(exp(x)*cos(x),x=0..3,12): sest:=evalf(%);
```

$$sest := -9.024261903$$

```
> int(exp(x)*cos(x),x=0..3); evalf(%);
```

$$\frac{1}{2}\mathbf{e}^{3}\cos(3)+\frac{1}{2}\mathbf{e}^{3}\sin(3)-\frac{1}{2}$$

$$-9.025029854$$

As we expected, Simpson's rule performs much better here.

Closing Remarks

We have barely scratched the surface. There are many other things Maple can do. Try exploring the help facility!

```
> ifactor(111111111111111111);
```

$$(3)^2\ (7)\ (11)\ (13)\ (19)\ (37)\ (333667)\ (52579)$$

```
> isprime(333667);
```

$$true$$

```
> isprime(333613);
```

$$false$$

```
> limit((exp(x)-1-x)/x^2,x=0);
```

$$\frac{1}{2}$$

```
> sum(k^4,k=1..n);
```

$$\frac{1}{5}(n+1)^5-\frac{1}{2}(n+1)^4+\frac{1}{3}(n+1)^3-\frac{1}{30}n-\frac{1}{30}$$

```
> sum(k^(-2),k=1..infinity);
```

$$\frac{1}{6}\pi^2$$

```
> fsolve(tan(x)=3*x,x,avoid={x=0},0..1.4);
```

$$1.324194450$$

```
> fsolve(tan(x)=3*x,x,1.4..5);
```

$$4.640683631$$

```
> solve({x+2*y=3,3*x-2*y=5},{x,y});
```

$$\{x=2, y=\frac{1}{2}\}$$

Experiment!

```
>
```